

Another model, which we will call PF-1, is now introduced. This model has several advantages over PF-0, including the following:

1. PF-1, in its simplest form, is the single-loop model shown in Fig. 4.3.5, but arteriovenous pathways (such as capillary beds in kidneys, skin, fat, brain, muscle, etc.) may easily be included (see Fig. 6.3.1).

2. The venous segments of the model, both pulmonary and systemic, have a structure much like that used in the artery in the left ventricular model (Fig. 4.2.5). This structure has, in very simple form, some tapering of impedance that tends to match the vessels near the heart (of impedance about 120 fluid ohms) to the higher resistive impedance of the capillary beds.

3. PF-1 is set up in algebraic form, so that changes can be made in any parameters at run time. The compliances are regarded as linear but with unstressed volumes (see Fig. 4.1.2). Thus a typical equation at a node in the equivalent lumped circuit model is

$$q_n = \int_0^t (f_{in} - f_{out})dt + q_n(0)$$

where $q_n(0) = q_{nu}(0) + q_{ns}(0)$ (the initial volume equals the sum of the initial unstressed and stressed volumes).

Pressure is obtained from stressed volume, using

$$p_n = q_{ns}/c_n = (q_n - q_{nu})/C_n$$

All quantities in this program are in CGS units, so the pressures must be converted to medical units (mm Hg) somewhere in the DYNAMIC part of the program, using $PNM = PN / 1332.0$.

In the INITIAL part of the program the initial total volumes are determined from initial unstressed volumes and initial end-diastolic pressures, both assumed to be known, using $QNIC = QNUIC + PNEDIC*$




Figure 4.3.5. A single-loop CV model, PF-1. Additional systemic arterial pathways may be added in parallel with R_{A3} .

$CN \cdot 1332$, for the n -th segment, where the pressure is given in mm Hg, and the model is to start at the beginning of systole.

4. The spaced half-sine wave used to produce the ventricular activity signal is the same as that used in the later ventricular models (see Fig. 4.2.4), but with a second harmonic sine wave added to shape the activity function more realistically (Snyder-68). Note, too, that this program is convenient for including a baroreceptor loop, as will be done in Section 4.4.

The following program for PF-1 contains coding that permits a myo-

cardial infarction (of one or both ventricles) to be introduced at a chosen time. Other defects, such as a VSD, atrial septal defect (ASD), or defective heart valves, may easily be added. The model contains a means for infusing blood or other fluids into the arterial system (flow FIS into QA1 at chosen time TIS). Unstressed volume, QU, is determined in INITIAL, together with initial stressed volume, QS. If infusion or bleeding (FIS negative) is made to occur, the stressed volume QST may be determined as a function of time as given in DYNAMIC. Also, unstressed volumes may be varied with the aid of simple additions to this model, to simulate some aspects of shock, for example.

PROGRAM PF-1
INITIAL

```
'Some constants and calculations of initial volumes'
Constant QP1U=7.8, PP1EDM=7.2, CP1=.0001
QP1IC= QP1U+ PP1EDM*CP1*1332.      $ 'Pulm. Art. 1'
Constant QP2U=23.4, PP2EDM= 7.0, CP2=.0003
QP2IC= QP2U + PP2EDM*CP2*1332.      $ 'Pulm. Art. 2'
Constant QP3U=210.5, PP3EDM= 6.6, CP3=.0027
QP3IC= QP3U+ PP3EDM*CP3*1332.      $ 'Pulm. Art. 3'
Constant QL1U=69., PL1EDM=4.45, CL1=.001
QL1IC= QL1U + PL1EDM*CL1*1332.      $ 'Pulm.Veins 1'
Constant QL2U=69., PL2EDM=3.62, CL2=.001
QL2IC= QL2U+ PL2EDM*CL2*1332.      $ 'Pulm.Veins 2'
Constant QLAU=814.5, PLAEDM=3.45, CLA=.01176
QLAIC= QLAU + PLAEDM*CLA*1332.      $ 'L. Atrium'
Constant QLVU=10., PLVEDM=4.0, LD=45.
QLVIC= QLVU+ PLVEDM*1332./LD        $ 'L. Ventricle'
Constant QA1U=35.1, PA1EDM=64.3, CA1=.00018
QA1IC= QA1U + PA1EDM*CA1*1332.      $ 'Syst. Art. 1'
Constant QA2U= 85., PA2EDM=64., CA2=.00023
QA2IC= QA2U+ PA2EDM*CA2*1332.      $ 'Syst. Art. 2'
Constant QA3U=710., PA3EDM=63., CA3=.00182
QA3IC= QA3U + PA3EDM*CA3*1332.      $ 'Syst. Art. 3'
Constant QV1U=909., PV1EDM=13.5, CV1=.021
QV1IC= QV1U+ PV1EDM*CV1*1332.      $ 'Syst. Veins 1'
Constant QV2U=1948., PV2EDM=7.2, CV2=.045
QV2IC= QV2U + PV2EDM*CV2*1332.      $ 'Syst. Veins 2'
Constant QRAU=1948., PRAEDM=6.64, CRA=.045
QRAIC= QRAU+ PRAEDM*CRA*1332.      $ 'Rt. Atrium'
Constant QRVU=10., PRVEDM=7.4, RD=72.
QRVIC= QRVU + PRVEDM*1332./RD      $ 'Rt. Ventricle'

'Calc. of total initial blood vol. QT, total un-
stressed volume QU, and stressed volume, QS at T=0.0'
QT=QP1IC+ QP2IC+ QP3IC+QL1IC+QL2IC+ QLAIC+ QLVIC
+QA1IC+ QA1IC+ QA3IC+ QV1IC+ QV2IC+ QRAIC+ QRVIC
```

```

QU=QP1U+ QP2U+ QP3U+ QL1U+ QL2U+ QLAU+ QLVU+ QA1U . . .
+QA2U+ QA3U+ QV1U+ QV2U+ QRAU+ QRVU
QS= QT- QU

END $'of initial'

DYNAMIC
    CONSTANT TF = 8.
TERMT(T .GE. TF)
Cinterval CINT=.02

DERIVATIVE
    Algorithm IALG = 4      $ '2nd order RK'
    Maxterval MAXT = .002 $ Nsteps NSTP = 1

    Constant THI=800., LSI=2500., RSI=350., DLS=0., DRS=0.
    'DLS and DRS are values of sudden changes in LS, RS at THI'
    'These changes should be negative for an infarct'
    LS= LSI + FCNSW(TI,0.,0.,DLS)
    RS= RSI + FCNSW(TI,0.,0.,DRS)
    TI=T - THI      $ ' THI is time of infarct'

    Constant TSA=0.1, TS=.3, TH=.8, PI=3.1416, KB=1., SV1=.9, SV2=.25
    LOGICAL XX
    X=T-ZOH(T,0.0,0.0,TH)
    XX=X .LE. TS
    STW=RSW(XX,X,0.0)
    SSW = SV1*SIN(PI*STW/TS)-SV2*SIN(2.*PI*STW/TS)
    ACTV=KB*BOUND(0.0,1.0,SSW)      $ 'Ventr. Pumping Activ.'

    'Pressure-Flow Equations start here'
    Constant RPW1=10., LP1=1.0, FP1IC=0.0, . . .
    KP1=1., RP1=10.      $ 'Pul. Art. 1'
    PP1= (QP1-QP1U)/CP1 +KP1 *RPW1*(FRV-FP1)
    FP1= INTEG((PP1-PP2-RP1*FP1)/LP1,FP1IC)
    OP1= INTEG(FRV-FP1,QP1IC) $ 'Note:CP1 and QP1IC in INITIAL'
    Constant RP2=40.      $ 'Pul. Art. 2'
    PP2= (QP2-QP2U)/CP2
    FP2= (PP2-PP3)/RP2
    QP2= INTEG(FP1-FP2,QP2IC)
    Constant RP3=80.      $ 'Pul. Art. 3'
    PP3= (QP3-QP3U)/CP3
    FP3= (PP3-PL1)/RP3
    QP3= INTEG(FP2-FP3,QP3IC)
    Constant RL1=30.      $ 'Pul. Vein 1'
    PL1= (QL1-QL1U)/CL1
    FL1= (PL1-PL2)/RL1
    QL1= INTEG(FP3-FL1,QL1IC)

```

```

Constant RL2=10., LL2=1.0, FL2IC=33.      $ 'Pul. Vein 2'
PL2= (QL2-QL2U)/CL2
FL2= INTEG((PL2-PLA-RL2*FL2)/LL2,FL2IC)
QL2= INTEG(FL1-FL2,QL2IC)
Constant RLA=5., LLA=1.0, FLAIC=0.0      $ 'Left Atrium'
PLA= (QLA-QLAU)/CLA
FLA= LIMINT((PLA-PLV-RLA*FLA)/LLA,FLAIC,0.0,1.E4)
QLA= INTEG(FL2-FLA,QLAIC)
Constant RLV=5., LLV=1., FLVIC=0.0      $ 'Left Ventr.'
SLV= LD*(1.-ACTV) + LS*ACTV
'LD given in INITIAL, LS in 6-th line of DERIVATIVE'
PLV= (QLV-QLVU)*SLV
FLV= LIMINT((PLV-PA1-RLV*FLV)/LLV,FLVIC,0.,1.E5)
QLV= INTEG(FLA-FLV,QLVIC)
Constant FA1IC=4.6, RA1=10., LA1=1.0      $ 'Aorta 1'
Constant RPW2=10., KP2=1.0, FIS=0.0, TIS=3.0
FI= FCNSW(T-TIS,0.0,0.0,FIS)      $ 'Infusion FI,T>TIS'
PA1= (QA1-QA1U)/CA1 +KP2*RPW2*(FLV-FA1)
FA1= INTEG((PA1-PA2-RA1*FA1)/LA1,FA1IC)
QA1= INTEG(FLV-FA1+FI,QA1IC)
Constant RA2=160.      $ 'Aorta 2'
PA2= (QA2-QA2U)/CA2
FA2= (PA2-PA3)/RA2
QA2= INTEG(FA1-FA2,QA2IC)
Constant RA3=1000.      $ 'System. Art.'
PA3= (QA3-QA3U)/CA3
FA3= (PA3-PV1)/RA3
QA3= INTEG(FA2-FA3,QA3IC)
Constant RV1=90.      $ 'System. Veins 1'
PV1= (QV1-QV1U)/CV1
FV1= (PV1-PV2)/RV1
QV1= INTEG(FA3-FV1,QV1IC)
Constant RV2=10., LV2=1., FV2IC=95.      $ 'System. Veins 2'
PV2= (QV2-QV2U)/CV2
FV2= INTEG((PV2-PRA-RV2*FV2)/LV2,FV2IC)
QV2= INTEG(FV1-FV2,QV2IC)
Constant RRA=5., LRA=1.0, FRAIC=0.      $ 'Rt. Atrium'
PRA= (QRA-QRAU)/CRA
FRA= LIMINT((PRA-PRV-RRA*FRA)/LRA, FRAIC, 0.,1.E4)
QRA= INTEG (FV2-FRA,QRAIC)
Constant RRV=5., LRV=1., FRVIC=6.0      $ 'Rt. Ventr.'
SRV= RD*(1.-ACTV) + RS*ACTV
PRV= (QRV-QRVU)*SRV
FRV= LIMINT((PRV-PP1-RRV*FRV)/LRV,FRVIC,0.,1.E5)
QRV= INTEG(FRA-FRV,QRVIC)

```

END \$'of Deriv.'

```

'Calculate Output Pressures in mmHg'
PP1M=PP1/1332. $ PP2M=PP2/1332. $ PP3M=PP3/1332.
PL1M=PL1/1332. $ PL2M=PL2/1332. $ PLAM=PLA/1332.
PLVM=PLV/1332. $ PA1M=PA1/1332. $ PA2M=PA2/1332.
PA3M=PA3/1332. $ PV1M=PV1/1332. $ PV2M=PV2/1332.
PRAM=PRA/1332. $ PRVM=PRV/1332.

```

```

'Find Total Volume and Stressed Volume as Functions of Time'
QTOT=QP1+QP2+QP3+QL1+QL2+QLA+QLV+QA1+QA2+QA3+QV1+QV2+QRA+QVR
QST = QTOT - QU $'Total Stressed Vol. as Fcn. of Time'

```

```

END $ 'of Dynamic'
END $ 'of Program'

```

Running normally, this program has a ventricular activity function and ventricular volumes, as shown in Fig. 4.3.6. The ventricular activity function has a somewhat slower rise and a faster fall as a result of the inclusion of a second harmonic sinusoidal term, but might be improved by clipping its rather sharp peak. The ventricular volume curves are much like those for PF-0 (see Fig. 4.3.2), but stroke volume is small at a value of 64 ml; since the heart period is fixed at 0.8 sec, this corresponds to a cardiac output of $64/0.8 = 80$ ml/s, which is slightly low for an adult male. The cardiac output will increase as blood volume is increased. This may be accomplished by including a factor M in all calculations for stressed volumes; M should ordinarily be set to unity but may be set to a larger value at run time to increase blood volume and cardiac output.

Other outputs for PF-1 are shown in Fig. 4.3.7. Here the right ventricular pressure PRVM (in medical units) is shown in part a, together with the pulmonary artery pressure, PP1M; the corresponding left side pressures, PLVM and PA1M, are shown in part b. The flow FLA into the left ventricle is shown in part c, together with the flow FLV into the first segment of aorta; note that FLA has an oscillatory decay after the aortic valve closes. In a better, more detailed model this oscillation may go negative on its downswings. The important left ventricular pressure versus volume locus in Fig. 4.3.7d shows the peak ventricular pressure, stroke volume VS, and ejection fraction (here about 50 percent).

The means for sudden introduction of weakened ventricles (myocardial infarction) are included in model PF-1. This is achieved by modifying the two commands used in the first part of the Derivative section of PF-1 to determine the ventricular systolic stiffnesses, LS and RS. These are normally set equal to the given constants, LSI and RSI, and function switches are included to change these values by DLS and DRS at time THI. Thus, if we set $DLS = -625.0$ and $DRS = -87.5$ at run time, a 25 percent weakening of systolic stiffness in each ventricle will occur at




Figure 4.3.6. (a) Ventricular activity function, ACTV, in model PF-1. (b) Ventricular volumes in model PF-1.

time THI—somewhat unrealistically in that the weakening in cardiac muscle will ordinarily tend to occur more slowly.

This model was set to run with DLS and DRS set to the negative values given above, and $THI = 2.4$; some outputs for this case are shown in Fig. 4.3.8. Here the sudden decrease in ventricular stiffnesses causes a corresponding sudden decrease in stroke volume, as shown in Fig. 4.3.8b, but a slight recovery begins to occur as atrial average volumes and pressures increase, thus causing increased filling, which in turn (by Starling's law) increases ventricular output. Thus the circulatory system has some inherent stability, even without baroreceptor control, which would tend




Figure 4.3.7. Model PF-1 in normal operation.
 (a) Right ventricular and pulmonary artery pressures.
 (b) Left ventricular and aortic pressures.
 (c) Aortic flow, FLV and mitral valve inflow FLA, to ventricle.
 (d) Pressure-volume plot for the left ventricle.

to give further recovery in case of reduced ventricular strength (see Section 4.5). Note that many other effects not included in this model bear upon the problem being considered; for example, reduced output pressure can reduce coronary flow to the heart muscles, which in turn further reduces pressures and flows.

Program PF-1 also includes means for either fluid infusion or a fixed rate of bleeding from the systemic arteries, as discussed above.




Figure 4.3.8. Model PF-1 with a simulated sudden infarction.
 (a) Sudden decreases in SRV and SLV at $THI = 2.4$ sec.
 (b) Ventricular volumes. Note the slow recovery in stroke volume after the infarct.